Project 2: Koch Snowflake and Fractals

Douglas Meade, Ronda Sanders, and Xian Wu Department of Mathematics

Overview

The word "fractal" is often used in referring any object that is recursively constructed so that it appears similar at all scales of magnification. There are many examples of complex real-life phenomena, such as chaos, ferns, mountains, river networks, biological growth, that can be described and studied using fractals. In this lab and project, we will analyze and generate a classic fractal, the Koch snowflake, and its variations. While it is natural to use a computer to do recursive constructions, we will focus on applications of sequences and series in our study.

Koch Snowflake

Basic Construction: It starts with an equilateral triangle. A smaller equilateral triangle is then added to each of the three sides. It is done in such a way that the base of each new triangle is the middle one-third of each side of the original figure. This process is repeated again and again in each successive iteration. The Koch snowflake is the limit of this construction.

Finding Area and Perimeter: First notice that we are getting a sequence of flakes with more and more sides by the same basic construction: each side is replaced by four sides of one-third of its length at the next level. Therefore, as we started at level 0 with 3 sides, say, of length 1, there are $3 * 4^n$ sides of length $(1/3)^n$ at level n. That gives a formula

$$TotPerim_n = 3 * 4^n * (1/3)^n = 3 * (4/3)^n$$

for the perimeter of the flake at level n. This sequence diverges and the perimeter of the Koch snowflake is hence infinite. To get a formula for the area, notice that the new flake at level $n \ge 1$ is obtained by adding an equilateral triangle of the side length $(1/3)^n$ to each side of the previous flake. Therefore, the total new area added at level $n \ge 1$ is:

$$AreaAdd_n = 3 * 4^{n-1} * (1/3)^n * (1/3)^n * \sin(\pi/3)/2 = (3/16) * \sqrt{3} * (4/9)^n.$$

Since the area at level 0 is $\sqrt{3}/4$ and the above area is added at each level thereafter, we hence obtain the following series for the area of the flake at level $n \ge 1$:

$$TotArea_n = TotArea_{n-1} + (3/16) * \sqrt{3} * (4/9)^n = \sqrt{3}/4 + \sum_{k=1}^n (3/16) * \sqrt{3} * (4/9)^k.$$

Apply our knowledge in geometric series, we obtain that the Koch snowflake has a finite area of

$$\sqrt{3}/4 + (3/16) * \sqrt{3} * (4/9)/(1 - (4/9)) = 2 * \sqrt{3}/5 = 0.4 * \sqrt{3} \approx 0.6928203232.$$

Project 2:

Your project report should follow the guidelines set forth in the Project Report Guidelines on the lab website and is due by the date specified by your TA. It should cover the following:

- 1. Present details to show that you fully understand the given Koch snowflake construction and verify the formulas and limits presented using your knowledge in sequences and series. Have a discussion on what you have learned, such as some interesting properties of the Koch snowflake. For example, as we have computed, the Koch snowflake has a finite area but infinite perimeter. Now, imagining that you have a container with the Koch snowflake as its base and fill it up with some paint. That means one could paint an infinite area (the interior surface of the container) with a finite amount of paint! The Koch snowflake is also an example of a curve that is everywhere continuous but nowhere differentiable.
- 2. Repeat the same discussion and analysis of at least two different variations of the Koch snowflake. Use your imagination but one of them must be constructed as follows: Instead of using triangles, start with a square and recursively add smaller squares. Your focus should be on how to use sequences and series to get formulas and limits of perimeters and areas. Included Maple worksheet should be used as a tool to help you to verify and visualize. Your TA will help you to modify the worksheet to generate the first few levels of your constructions.

Spring 2008 Created 3/28/2008

Koch Snowflake Construction in Maple:

1. Note: You should always restart from the beginning after any modification.

```
> restart;
> with(plots):
```

2. **Initial data and results (level 0):** Each side of the initial equilateral triangle is determined by two of its endpoints (vertices) so we need a sequence of four points with the first and the last being the same vertex. Notice that plot automatically connects a given sequence of points.

```
> A[1]:=[0,0];
> A[2]:=[1/2,sin(Pi/3)];
> A[3]:=[1,0];
> A[4]:=A[1];
> LenSide[0]:=1;
> TotArea[0]:=LenSide[0]*sin(Pi/3)*LenSide[0]/2;
> NumSide[0]:=3;
> TotPerim[0]:=NumSide[0]*LenSide[0];
> plot([A[1],A[2],A[3],A[4]],axes=NONE,scaling=constrained, title=sprintf("Koch snowflake (level 0) \n area = % 8.5f, perimeter=%8.5f,sides=%d",TotArea[0],TotPerim[0],NumSide[0]));
```

3. **Results up to level n=20:** Please pay close attention as your TA works out recursive and general formulas for TotPerim[n] (a sequence) and TotArea[n] (a series) by hand.

```
> n:=20:
> for k from 1 to n do
> level:=k;
> LenSide[k]:=LenSide[k-1]/3.;
> TotArea[k]:=TotArea[k-1]+NumSide[k-1]*TotArea[0]/9.^k;
> NumSide[k]:=4*NumSide[k-1];
> TotPerim[k]:=NumSide[k]*LenSide[k];
> end do;
```

4. Construct sequences of snowflake vertices and plots: Between each pair of adjacent vertices A[m] and A[m+1], we need to add three new vertices B[m], E[m], and C[m]. (Note: Letter D is reserved in Maple.) We then have to re-index them to form a new sequence of vertices A[m] at the next level. Computations for levels above 5 may require too much time and memory for your computer to handle.

```
> n:=5: # For program testing, use n=2 or 3.
> for k from 1 to n do
    for m from 1 to 3*4^(k-1) do
      B[m] := A[m] + (A[m+1] - A[m])/3;
      E[m] := A[m] + (A[m+1] - A[m])/2 + sqrt(3)/6.*[A[m] [2] - A[m+1] [2], A[m+1] [1] - A[m] [1]];
      C[m] := A[m] + 2*(A[m+1] - A[m])/3;
    end do;
    for m from 1 to 3*4^(k-1) do
      Temp[4*(m-1)+1]:=simplify(A[m]);
      Temp[4*(m-1)+2]:=simplify(B[m]);
      Temp[4*(m-1)+3]:=simplify(E[m]);
      Temp[4*(m-1)+4]:=simplify(C[m]);
    end do;
    for m from 1 to 3*4^k do
     A[m] := Temp[m];
    end do;
    A[3*4^k+1] := A[1];
    SFplots[k] := seq(plot([A[m],A[m+1]]),m=1..3*4^k):
> end do:
> for k from 1 to n do
> display([SFplots[k]],axes=NONE,scaling=constrained,title=sprintf("Koch snowflake(level %a)
\n area = %8.5f,perimeter= %8.5f,sides = %d",k,TotArea[k],TotPerim[k],NumSide[k]));
> end do;
```

Spring 2008 Created 3/28/2008